Dynamics of interacting particles

Abhishek Sharma
BioMIP
Ruhr Universität, Bochum, Germany

The interactions which arise from two dimensional particles can be assumed as superpostion of interaction from individual electrode . In the simplest form, assuming each electrode as a point, interaction from each electrode can be approximated by Yukawa type potential. The potential is monotone increasing, implying that force is always attractive (Currently, I don’t know the correct parameters to desribe the physically realistic interaction). Mathematically, this potential is given by :

$$V_{\text{Yukawa}}(r) = -g^2 \frac{e^{-k\,r}}{r}$$

where, g and k are scaling constants. So, we can define the Yukawa Potential as :

```
ClearAll["Global`*"]

VYukawa[{p_, q_}, r_] := -p^2 (e^(-q r))/r

p1 = Plot[Evaluate@Table[VYukawa[{1, q}, r], {q, 1, 20, 4}], {r, 0, 1},
    PlotLabel → "Yukawa Potential v/s distance", FrameLabel → {"r", "Potential"},
    BaseStyle → {FontWeight → "Bold", FontSize → 10}, Frame → True,
    PlotLegends → Placed[{"k=1", "k=5", "k=9", "k=13", "k=17"}, {Right, Bottom }]];

p2 = Plot[Evaluate[Table[VYukawa[{g, 1}, r], {g, 1, 20, 4}]], {r, 0, 1},
    PlotLegends → Placed[{"g=1", "g=5", "g=9", "g=13", "g=17"}, {Right, Bottom }],
    PlotLabel → "Yukawa Potential v/s distance", FrameLabel → {"r", "Potential"},
    BaseStyle → {FontWeight → "Bold", FontSize → 10}, Frame → True];
```

```
GraphicsRow[{p1, p2}]
```



Extension to higher dimensions for particles and interacting surface

It is easy to extend these dimensions to higher dimensions. We extend this formulation to finite number of patterns on docking station (upto 9 docklets) and for single lablets to study their dynamics.

Case 1 : Creation of docking station surface potential

Suppose there exists two particles, which interact with each other via Yukawa Potential. The pair interaction force can be calculated by the gradient of the interaction potential. The pair interaction force in two dimensions is given by :

$$\texttt{Grad}\left[\texttt{VYukawa}\left[\{p, q\}, \sqrt{x^2+y^2}\right], \{x, y\}\right]$$

$$\left\{\frac{e^{-q\sqrt{x^2+y^2}}\, p^2\, x}{\left(x^2+y^2\right)^{3/2}}+\frac{e^{-q\sqrt{x^2+y^2}}\, p^2\, q\, x}{x^2+y^2}, \frac{e^{-q\sqrt{x^2+y^2}}\, p^2\, y}{\left(x^2+y^2\right)^{3/2}}+\frac{e^{-q\sqrt{x^2+y^2}}\, p^2\, q\, y}{x^2+y^2}\right\}$$

In normalized system, the interaction force between two particles at (x_1, y_1) and (x_2, y_2) is given by :

$$\texttt{F1} = \texttt{FullSimplify}\left[\texttt{Grad}\left[\texttt{VYukawa}\left[\{p, q\}, \sqrt{(x2-x1)^2+(y2-y1)^2}\right], \{x1, y1\}\right]\right]$$

$$\left\{\frac{e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\, p^2\, (x1-x2)\left(1+q\sqrt{(x1-x2)^2+(y1-y2)^2}\right)}{\left((x1-x2)^2+(y1-y2)^2\right)^{3/2}}, \frac{e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\, p^2\left(1+q\sqrt{(x1-x2)^2+(y1-y2)^2}\right)(y1-y2)}{\left((x1-x2)^2+(y1-y2)^2\right)^{3/2}}\right\}$$

```
F2 = FullSimplify[Grad[VYukawa[{p, q}, Sqrt[(x2 - x1)^2 + (y2 - y1)^2]], {x2, y2}]]
```

$$\left\{-\frac{e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\,p^2\,(x1-x2)\left(1+q\sqrt{(x1-x2)^2+(y1-y2)^2}\right)}{\left((x1-x2)^2+(y1-y2)^2\right)^{3/2}},\; -\frac{e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\,p^2\left(1+q\sqrt{(x1-x2)^2+(y1-y2)^2}\right)(y1-y2)}{\left((x1-x2)^2+(y1-y2)^2\right)^{3/2}}\right\}$$

```
F1 == -F2
```

```
True
```

```
Curl[F1, {x, y}]
```

```
0
```

As, $F_1 = -F_2$ and Curl of the force is zero, this shows that the force is conservative.

Keeping the simple case in two dimensions, we define Lablet as a combination of eight electrodes. In two dimensional coodinate system, the position of the centre of each electrode is given by (in terms of scaled units):

{e1, e2, e3, e4, e5 ,e6, e7, e8} = {{-1, -1}, {0, -1}, {1, -1}, {-1, 0}, {1, 0}, {-1, 1}, {0, 1}, {1, 1}}

```
e1 = {-1, -1}; e2 = {0, -1}; e3 = {1, -1}; e4 = {-1, 0};
e5 = {1, 0}; e6 = {-1, 1}; e7 = {0, 1}; e8 = {1, 1};
```

```
pos = Transpose[{e1, e2, e3, e4, e5, e6, e7, e8}]
```

```
{{-1, 0, 1, -1, 1, -1, 0, 1}, {-1, -1, -1, 0, 0, 1, 1, 1}}
```

```
posAll = Flatten[Table[
    Table[{i, j}, {8}] + MapThread[{#1, #2} &, pos], {i, -3, 3, 3}, {j, -3, 3, 3}], 2];
```

```
Graphics[Disk[#, 0.2] & /@posAll]
```



In two dimensional coordinate system, we define the Yukawa Potential as :

$$\text{VYukawa}[\{p\_, q\_\}, \{x\_, y\_, z\_\}, \{x1\_, y1\_\}] := -p^2 \frac{e^{-q\sqrt{(x-x1)^2+(y-y1)^2+z^2}}}{\sqrt{(x-x1)^2+(y-y1)^2+z^2}}$$

```
Plot3D[VYukawa[{1, 0}, {x, y, 0}, {0., 0.}], {x, -1, 1}, {y, -1, 1}]
```



```
Pot[{p_, q_}, {x_, y_, z_}] = Total[MapThread[VYukawa[{p, q}, {x, y, z}, {#1, #2}] &, pos]]
```

$$-\frac{e^{-q\sqrt{(-1+x)^2+(-1+y)^2+z^2}}\,p^2}{\sqrt{(-1+x)^2+(-1+y)^2+z^2}}-\frac{e^{-q\sqrt{x^2+(-1+y)^2+z^2}}\,p^2}{\sqrt{x^2+(-1+y)^2+z^2}}-\frac{e^{-q\sqrt{(1+x)^2+(-1+y)^2+z^2}}\,p^2}{\sqrt{(1+x)^2+(-1+y)^2+z^2}}-\frac{e^{-q\sqrt{(-1+x)^2+y^2+z^2}}\,p^2}{\sqrt{(-1+x)^2+y^2+z^2}}-$$
$$\frac{e^{-q\sqrt{(1+x)^2+y^2+z^2}}\,p^2}{\sqrt{(1+x)^2+y^2+z^2}}-\frac{e^{-q\sqrt{(-1+x)^2+(1+y)^2+z^2}}\,p^2}{\sqrt{(-1+x)^2+(1+y)^2+z^2}}-\frac{e^{-q\sqrt{x^2+(1+y)^2+z^2}}\,p^2}{\sqrt{x^2+(1+y)^2+z^2}}-\frac{e^{-q\sqrt{(1+x)^2+(1+y)^2+z^2}}\,p^2}{\sqrt{(1+x)^2+(1+y)^2+z^2}}$$

```
Plot3D[Pot[{2, 2}, {x, y, 0}], {x, -3, 3}, {y, -3, 3}, MaxRecursion→3]
```



So, this could be easily extended to multiple copies, to create a region of docking station :

```
Pot[{p_, q_}, {x_, y_, z_}] =
    Total[MapThread[VYukawa[{p, q}, {x, y, z}, {#1, #2}] &, Transpose[posAll]]];
```

```
ContourPlot[Pot[{1, 0}, {x, y, 0}], {x, -6, 6},
  {y, -6, 6}, MaxRecursion→ 2, Contours → 8, ColorFunction→ Hue]
```



Equipotential surface around a particle in three dimensional space

Consider a free moving particle at a position (X, Y, Z) and orientation (θ, ϕ, ψ) with respect the three axes passing through the centre of the lablet. To get coordinates of all electrodes general direction defined by orientation angles (θ, ϕ, ψ), we first rotate the lablet containing electrodes at the origin, and translate it to the correct position.

The generalized rotation matrix along three axis is given by,

```
rotMat[θ_, ϕ_, ψ_] :=
  RotationMatrix[θ, {1, 0, 0}].RotationMatrix[ϕ, {0, 1, 0}].RotationMatrix[ψ, {0, 0, 1}]
```

```
rotMat[θ, ϕ, ψ] // MatrixForm
```

$$\begin{pmatrix} \text{Cos}[\phi]\,\text{Cos}[\psi] & -\text{Cos}[\phi]\,\text{Sin}[\psi] & \text{Sin}[\phi] \\ \text{Cos}[\psi]\,\text{Sin}[\theta]\,\text{Sin}[\phi]+\text{Cos}[\theta]\,\text{Sin}[\psi] & \text{Cos}[\theta]\,\text{Cos}[\psi]-\text{Sin}[\theta]\,\text{Sin}[\phi]\,\text{Sin}[\psi] & -\text{Cos}[\phi]\,\text{Sin}[\theta] \\ -\text{Cos}[\theta]\,\text{Cos}[\psi]\,\text{Sin}[\phi]+\text{Sin}[\theta]\,\text{Sin}[\psi] & \text{Cos}[\psi]\,\text{Sin}[\theta]+\text{Cos}[\theta]\,\text{Sin}[\phi]\,\text{Sin}[\psi] & \text{Cos}[\theta]\,\text{Cos}[\phi] \end{pmatrix}$$

The position of all the electrodes of the particle are given by,

```
e1 = {-1, -1, 0}; e2 = {0, -1, 0}; e3 = {1, -1, 0}; e4 = {-1, 0, 0};
e5 = {1, 0, 0}; e6 = {-1, 1, 0}; e7 = {0, 1, 0}; e8 = {1, 1, 0};
```

```
electrodes = {e1, e2, e3, e4, e5, e6, e7, e8};
```

So, rotating each electrode of the lablet

```
rotElectrodes[{θ_, ϕ_, ψ_}] := (rotMat[θ, ϕ, ψ].#) &/@ electrodes
```

```
posElectrodes[{x_, y_, z_}, {θ_, ϕ_, ψ_}] := (#+{x, y, z}) &/@rotElectrodes[{θ, ϕ, ψ}]
```

So, with this generalize function for particles, we can rotate the electrodes accordingly.

We can create generalized force field using Yukawa potential, for a particle as function of position and rotation.

```
VYukawa3D[{p_, q_}, {x_, y_, z_}, {x1_, y1_, z1_}] := -p^2 e^(-q Sqrt[(x-x1)^2+(y-y1)^2+(z-z1)^2]) / Sqrt[(x-x1)^2+(y-y1)^2+(z-z1)^2]
```

$$\texttt{VYukawa3D[\{p\_, q\_\}, \{x\_, y\_, z\_\}, \{x1\_, y1\_, z1\_\}] := } -p^2 \frac{e^{-q\sqrt{(x-x1)^2+(y-y1)^2+(z-z1)^2}}}{\sqrt{(x-x1)^2+(y-y1)^2+(z-z1)^2}}$$

And, generalized expression for the Lablet potential and force in three dimensions is given by :

```
VYukawa3D[{p_, q_}, {x_, y_, z_}, {x1_, y1_, z1_}, {θ_, ϕ_, ψ_}] :=
  Total[MapThread[VYukawa3D[{p, q}, {x, y, z}, {#1, #2, #3}] &,
      Transpose[posElectrodes[{x1, y1, z1}, {θ, ϕ, ψ}]]]]

Field3D[{p_, q_}, {x_, y_, z_}, {x1_, y1_, z1_}, {θ_, ϕ_, ψ_}] =
    Grad[VYukawa3D[{p, q}, {x, y, z}, {x1, y1, z1}, {θ, ϕ, ψ}], {x, y, z}];

{Curl[Field3D[{p, q}, {x2, y2, z2}, {x1, y1, z1}, {θ, ϕ, ψ}], {x1, y1, z1}],
  Curl[Field3D[{p, q}, {x2, y2, z2}, {x1, y1, z1}, {θ, ϕ, ψ}], {x2, y2, z2}]}
```

```
{{0, 0, 0}, {0, 0, 0}}
```

So, we can see equipotential surfaces around the Lablet :

```
ContourPlot3D[Evaluate[VYukawa3D[{3, 3}, {x, y, z}, {0., 0., 0.}, π/180 {15., 15., 15.}]],
  {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Mesh → None, Contours → 3, MaxRecursion→ 2,
  ContourStyle → Directive[Orange, Opacity[0.3], Specularity[White, 30]],
  PlotLabel → "Equipotential surface around a Lablet",
  BaseStyle → {FontWeight→ "Bold", FontSize→ 10}]
```



Pairwise interaction force between two Lablets

Consider two Lablets with their COG at position vectors $(X_1,\ Y_1,\ Z_1)$ and $(X_2,\ Y_2,\ Z_2)$, and respective orientations given by $(\theta_1,\ \phi_1,\ \psi_1)$ and $(\theta_2,\ \phi_2,\ \psi_2)$.
As the force between two particles acts only along the line joining their center of masses, the total force can be written as the summation of forces from individual electrodes.

```
pairForce[{p_, q_}, {x1_, y1_, z1_}, {θ1_, ϕ1_, ψ1_}, {x2_, y2_, z2_}, {θ2_, ϕ2_, ψ2_}] =
    Total[MapThread[Field3D[{p, q}, {#1, #2, #3}, {x1, y1, z1}, {θ1, ϕ1, ψ1}] &,
        Transpose[posElectrodes[{x2, y2, z2}, {θ2, ϕ2, ψ2}]]]];
```

```
{Curl[pairForce[{p, q}, {x1, y1, z1}, {θ1, ϕ1, ψ1}, {x2, y2, z2}, {θ2, ϕ2, ψ2}],
    {x1, y1, z1}], Curl[pairForce[{p, q}, {x1, y1, z1},
      {θ1, ϕ1, ψ1}, {x2, y2, z2}, {θ2, ϕ2, ψ2}], {x2, y2, z2}]}
```

```
{{0, 0, 0}, {0, 0, 0}}
```

As the mathematical expressions becomes too long, we can check the forces by putting position and orientation vectors, as an example putting arbitrary values of positional and orientational vectors

```
F1 = pairForce[{1., 1.}, {1., 2., 3.}, {4., 5., 6.}, {6., 5., 4.}, {3., 2., 1.}];
F2 = pairForce[{1., 1.}, {6., 5., 4.}, {3., 2., 1.}, {1., 2., 3.}, {4., 5., 6.}];
```

```
F1 == -F2
```

```
True
```

```
F1
```

```
{0.0323819, 0.0140289, 0.00743171}
```

To calculate torque between two Lablets, we consider two particles with position and orientational vectors as $(X_1, Y_1, Z_1, \theta_1, \phi_1, \psi_1)$ and $(X_2, Y_2, Z_2, \theta_2, \phi_2, \psi_2)$. We measure the torque on the particles at the centre of gravity.

So, with respect to COG of the particles, the position vector of electrodes are given by :

For some reasons, *Mathematica* inbulit function Cross takes too much of time while evalulation so I define myself crossproduct function as :

```
crossproduct[{x1_, y1_, z1_}, {x2_, y2_, z2_}] := {-y2 z1 + y1 z2, x1 z1 - x1 z2, -x1 y1 + x1 y2}
```

```
pairTorque[{p_, q_}, {x1_, y1_, z1_}, {θ1_, ϕ1_, ψ1_}, {x2_, y2_, z2_}, {θ2_, ϕ2_, ψ2_}] =
    Total[MapThread[crossproduct[{#1, #2, #3},
            Field3D[{p, q}, {#1, #2, #3}, {x1, y1, z1}, {θ1, ϕ1, ψ1}]] &,
        Transpose[posElectrodes[{x2, y2, z2}, {θ2, ϕ2, ψ2}]]]];
```

As an example,

```
pairTorque[{1., 1.}, {0., 0., 0.}, {0., 0., 90 π/180}, {5., 0., 0.}, {0., 0., 90 π/180}]
```

$\{0., 0., -6.93889\times10^{-18}\}$

Based on these type of interactions, we can minimize the energy of the particles to calculate equilibriated self-assembled structures.

In this Mathematica file, we will consider various types of interactions that could be useful for self - assembly of particles. We will compare the order of interaction potential energies and corresponding forces, that could be useful particle - particle interactions at small and large length scales. Basically, there are always some interactions that could not be avoided for example gravitational forces, drag force etc. Usually, we have to compete with these interactions to find interactions for self - assenmbly. In this file, we will first calculate each type of interaction and give a comparitive study.

Interactions between moving Lablets

Understanding dynamics of Lablets in Yukawa type field

```
VYukawa[{p_, q_}, r_] := p^2 e^(-q r)/r
```

```
sol = NDSolve[{D[x[t], {t, 2}] == (D[VYukawa[{1, 1}, x], x] /. x → x[t]),
      x'[0] == 0, x[0] == 5}, x[t], {t, 0, 45}]
```

```
{{x[t] → InterpolatingFunction[ Domain: {{0., 45.}} Output: scalar ][t]}}
```

```
Plot[x[t] /. sol, {t, 0, 45}, PlotStyle → {Thick, Blue},
  AxesLabel → {"Time ", "Gap between Lablets"}]
```



Extension to two dimensions

In this subsection, we extend dynamics of particles in Yukawa field to two dimensions. We consider two dimensional analogue of docking station, which includes one dimensional array of electrodes, whole interaction potential is defined by Yukawa type potential given by :

$$V_{\text{Yukawa}} = \frac{p^2}{\sqrt{(x-x_0)^2+(y-y_0)^2}} e^{-q\sqrt{(x-x_0)^2+(y-y_0)^2}}$$

where, $\{x_0, y_0\}$ defines the position of the electrode on docking station, $\{x, y\}$ describes the position of the electrode on the Lablet. For a simple calculation, inititally, we consider Lablet as a single electrode, hence we define the potential from docking station as :

```
VYukawa[{p_, q_, n_}, {x_, y_}, {x0_, y0_}] := (-1)^n p^2 e^(-q Sqrt[(x-x0)^2 + (y-y0)^2]) / Sqrt[(x-x0)^2 + (y-y0)^2]
```

$$\text{VYukawa}[\{p\_, q\_, n\_\}, \{x\_, y\_\}, \{x0\_, y0\_\}] := (-1)^n p^2 \frac{e^{-q\sqrt{(x-x0)^2+(y-y0)^2}}}{\sqrt{(x-x0)^2+(y-y0)^2}}$$

We define periodic dimensionaless position of docking electrodes :

```
nE = 10; (* Number  of electrodes *)
```

```
positions= {#, 0} & /@Range[-nE/2, nE/2]
```

```
{{-5, 0}, {-4, 0}, {-3, 0}, {-2, 0}, {-1, 0}, {0, 0}, {1, 0}, {2, 0}, {3, 0}, {4, 0}, {5, 0}}
```

```
nType = {0, 1, 0, 1, 0, 1, 0, 1, 0, 1, 0};
```

```
VYukawaTot[x_, y_] =
  Total@MapThread[VYukawa[{1, 1, #1}, {x, y}, #2] &, {nType, positions}]
```

$$\frac{e^{-\sqrt{(-5+x)^2+y^2}}}{\sqrt{(-5+x)^2+y^2}}-\frac{e^{-\sqrt{(-4+x)^2+y^2}}}{\sqrt{(-4+x)^2+y^2}}+\frac{e^{-\sqrt{(-3+x)^2+y^2}}}{\sqrt{(-3+x)^2+y^2}}-\frac{e^{-\sqrt{(-2+x)^2+y^2}}}{\sqrt{(-2+x)^2+y^2}}+\frac{e^{-\sqrt{(-1+x)^2+y^2}}}{\sqrt{(-1+x)^2+y^2}}-\frac{e^{-\sqrt{x^2+y^2}}}{\sqrt{x^2+y^2}}+\frac{e^{-\sqrt{(1+x)^2+y^2}}}{\sqrt{(1+x)^2+y^2}}-\frac{e^{-\sqrt{(2+x)^2+y^2}}}{\sqrt{(2+x)^2+y^2}}+\frac{e^{-\sqrt{(3+x)^2+y^2}}}{\sqrt{(3+x)^2+y^2}}-\frac{e^{-\sqrt{(4+x)^2+y^2}}}{\sqrt{(4+x)^2+y^2}}+\frac{e^{-\sqrt{(5+x)^2+y^2}}}{\sqrt{(5+x)^2+y^2}}$$

```
ContourPlot[VYukawaTot[x, y], {x, -4, 4}, {y, 0, 2}, MaxRecursion→ 5]
```



Force on a single electrode type Lablet is given by gradient of potential, which is equal to

```
ForceLablet[x_, y_] = Grad[VYukawaTot[x, y], {x, y}];
```

```
StreamPlot [ForceLablet[x, y], {x, -4, 4},
  {y, 0, 4}, StreamScale → Large, StreamPoints → Fine]
```




```
{Fx, Fy} = ForceLablet[x, y];

eqns = {(Fx /. {x → x[t], y → y[t]}) == D[x[t], {t, 2}], (Fy /. {x → x[t], y → y[t]}) ==
       D[y[t], {t, 2}], x'[0] == 0, y'[0] == 0, x[0] == -4, y[0] == 0.5};

sol = NDSolve[eqns, {x[t], y[t]}, {t, 0, 10}]

{{x[t] → InterpolatingFunction[ Domain: {{0., 10.}} Output: scalar ][t],
  y[t] → InterpolatingFunction[ Domain: {{0., 10.}} Output: scalar ][t]}}

ParametricPlot [{x[t], y[t]} /. sol, {t, 0, 10}, PlotStyle → {Blue, Thick}];
```

Interactions between two Lablets

Case 1 : Each Lablet is defined by a single electrode

Two dimensional formulation of Yukawa potential is given by :

$$\texttt{VYukawa[\{p\_, q\_, n\_\}, \{x\_, y\_\}, \{x0\_, y0\_\}] :=} \ (-1)^n p^2 \frac{e^{-q\sqrt{(x-x0)^2+(y-y0)^2}}}{\sqrt{(x-x0)^2+(y-y0)^2}}$$

As, each Lablet is defined as a single electrode, we consider Lablet as a point particle, hence the interaction force between two such point particles in two dimensions is given by :

```
PairIntForce[{p_, q_, n_}, {x1_, y1_}, {x2_, y2_}] =
  Grad[VYukawa[{p, q, n}, {x1, y1}, {x2, y2}], {x1, y1}]
```

$$\left\{\frac{(-1)^{1+n} e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\, p^2\,(x1-x2)}{\left((x1-x2)^2+(y1-y2)^2\right)^{3/2}} + \left((-1)^{1+n} e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\, p^2\, q\,(x1-x2)\right)\Big/\left((x1-x2)^2+(y1-y2)^2\right),\right.$$
$$\left.\frac{(-1)^{1+n} e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\, p^2\,(y1-y2)}{\left((x1-x2)^2+(y1-y2)^2\right)^{3/2}} + \left((-1)^{1+n} e^{-q\sqrt{(x1-x2)^2+(y1-y2)^2}}\, p^2\, q\,(y1-y2)\right)\Big/\left((x1-x2)^2+(y1-y2)^2\right)\right\}$$

```
FullSimplify[PairIntForce[{p, q, n}, {x1, y1}, {x2, y2}]] ==
  -FullSimplify[PairIntForce[{p, q, n}, {x2, y2}, {x1, y1}]]
```

```
True
```

```
rules = {x1 → x1[t], x2 → x2[t], y1 → y1[t], y2 → y2[t]};
```

```
{Fxp1, Fyp1} = FullSimplify[PairIntForce[{p, q, n}, {x1, y1}, {x2, y2}]] /. rules;
```

```
{Fxp2, Fyp2} = FullSimplify[PairIntForce[{p, q, n}, {x2, y2}, {x1, y1}]] /. rules;
```

Hence, using this pair interaction force, we can formulate the equations of motion for the two particles as :

```
govEquations[{m1_ , m2_ }, {p_, q_, n_}] =
    {Fxp1 == m1  x1''[t], Fyp1 == m1  y1''[t], Fxp2 == m1  x2''[t], Fyp2 == m2  y2''[t]};
```

```
initVelocities = {x1'[0] == 0.1, y1'[0] == 0, x2'[0] == 0, y2'[0] == 0};
```

```
initPositions = {x1[0] == 1, y1[0] == 1, x2[0] == 5, y2[0] == 6};
```

```
initConditions = {initVelocities, initPositions} // Flatten;
```

```
totalEquations[{m1_ , m2_ }, {p_, q_, n_}] =
    {govEquations[{m1 , m2 }, {p, q, n}], initConditions} // Flatten;
```

```
sol = NDSolve[totalEquations[{1, 1}, {5, 1, 0}], {x1[t], y1[t], x2[t], y2[t]}, {t, 0, 200}]
```

```
{{x1[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t],
  y1[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t],
  x2[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t],
  y2[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t]}}
```

```
({x1[t], y1[t]} /. sol)
```

```
{{InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t],
  InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t]}}
```

```
Show[ParametricPlot [({x1[t], y1[t]} /. sol), {t, 0, 200}, PlotStyle → {Blue, Thick}],
  ParametricPlot [({x2[t], y2[t]} /. sol), {t, 0, 200}, PlotStyle → {Red, Thick}],
  PlotRange → All]
```



Case 2 : Each Lablet is defined by combination of electrodes : Pure Translational Motion

Here, we define the two dimensional analogue of a Lablets, with four electrode at the corners, whose interactions are described by Yukawa potential. Lablet is defined as a unit square, whose positions of electrodes wrt. center of Lablet are given by :

```
Clear[positions]
```

```
VYukawa[{p_, q_, n_}, {x_, y_}, {x0_, y0_}] := (-1)^n p^2 E^(-q Sqrt[(x-x0)^2 + (y-y0)^2]) / Sqrt[(x-x0)^2 + (y-y0)^2]

positions[x_, y_] := {{-1/2 + x, -1/2 + y}, {-1/2 + x, 1/2 + y}, {1/2 + x, -1/2 + y}, {1/2 + x, 1/2 + y}};

genPositions[x_, y_, θ_] :=
  {x, y} + # & /@ (RotationMatrix[θ].# & /@ {{-1/2, -1/2}, {-1/2, 1/2}, {1/2, 1/2}, {1/2, -1/2}})

nType = {1, 1, 1, 1};

VYukawaTot[{x_, y_}, {x0_, y0_, θ0_}] = Total@
    (MapThread[VYukawa[{1, 1, #1}, {x, y}, #2] &, {nType, genPositions[x0, y0, θ0]}]);

Plot3D[VYukawaTot[{x, y}, {0, 0, 0}], {x, -1, 1}, {y, -1, 1}]
```



Now, we consider two different Lablets, in different orientations at whose positions and angles (wrt. X axis) are given by :
$P_1 = \{x_1, y_1, \theta_1\}$, $P_2 = \{x_2, y_2, \theta_2\}$. Suppose initital positions of Lablets are : {{-1,-1}, {6,5}}

```
FieldLablet[{x_, y_}, {x0_, y0_, θ0_}] = Grad[VYukawaTot[{x, y}, {x0, y0, θ0}], {x, y}];

pairForce[{x1_, y1_, θ1_}, {x2_, y2_, θ2_}] =
   Total@ (FieldLablet[#, {x1, y1, θ1}] & /@genPositions[x2, y2, θ2]);

{pairForce[{-1, -1, 3π/4}, {5, 6, 2π/3}], pairForce[{5, 6, 2π/3}, {-1, -1, 3π/4}]} // N
```

```
{{0.000162891, 0.00018942}, {-0.000162891, -0.00018942}}
```

```
{Curl[pairForce[{x1, y1, θ1}, {x2, y2, θ2}], {x1, y1}],
  Curl[pairForce[{x1, y1, θ1}, {x2, y2, θ2}], {x2, y2}]}
```

```
{0, 0}
```

Based on the pair interaction force, we can write the governing equations of motion as we did in the last section.

```
rules = {x1 → x1[t], x2 → x2[t], y1 → y1[t], y2 → y2[t]};
```

```
θfixedRules = {θ1 → 0, θ2 → 0};
```

```
{Fxp1, Fyp1} = pairForce[{x1, y1, θ1}, {x2, y2, θ2}] /. rules /. θfixedRules;
```

```
{Fxp2, Fyp2} = pairForce[{x2, y2, θ2}, {x1, y1, θ1}] /. rules /. θfixedRules;
```

Hence, using this pair interaction force, we can formulate the equations of motion for the two particles as :

```
govEquations[{m1_ , m2_ }, {p_, q_, n_}] =
    {Fxp1 == m1 x1''[t], Fyp1 == m1 y1''[t], Fxp2 == m1 x2''[t], Fyp2 == m2 y2''[t]};
```

```
initVelocities = {x1'[0] == 0.1, y1'[0] == 0, x2'[0] == 0, y2'[0] == 0};
```

```
initPositions = {x1[0] == 1, y1[0] == 1, x2[0] == 5, y2[0] == 6};
```

```
initConditions = {initVelocities, initPositions} // Flatten;
```

```
totalEquations[{m1_ , m2_ }] =
    {govEquations[{m1 , m2 }, {p, q, n}], initConditions} // Flatten;
```

```
totalEquations[{1, 1}] // MatrixForm ;
```

```
sol = NDSolve[totalEquations[{1, 1}], {x1[t], y1[t], x2[t], y2[t]}, {t, 0, 200}]
```

```
{{x1[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t],
    y1[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t],
    x2[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t],
    y2[t] → InterpolatingFunction[ Domain: {{0., 200.}} Output: scalar ][t]}}
```

```
Show[ParametricPlot [({x1[t], y1[t]} /. sol), {t, 0, 200}, PlotStyle → {Blue, Thick}],
  ParametricPlot [({x2[t], y2[t]} /. sol), {t, 0, 200}, PlotStyle → {Red, Thick}],
  PlotRange → All]
```



Case 2 : Coupled Rotational and Translational Motion

In case if coupled rotational and translational motion, we consider pair force interaction and pair torques between two particles. Pair force is defined in the last subsection, here we defined both pair force and pair torque, and then write down the equations of both rotational and translational motion.

```
VYukawaTot[{x_, y_}, {x0_, y0_, θ0_}] =
    Total@(MapThread[VYukawa[{0.7, 2, #1}, {x, y}, #2] &,
          {nType, genPositions[x0, y0, θ0]}]);
```

```
FieldLablet[{x_, y_}, {x0_, y0_, θ0_}] = Grad[VYukawaTot[{x, y}, {x0, y0, θ0}], {x, y}];
```

```
pairForce[{x1_, y1_, θ1_}, {x2_, y2_, θ2_}] =
    Total@(FieldLablet[#, {x1, y1, θ1}] & /@genPositions[x2, y2, θ2]);
```

To calculate the pair torque, we calculate individual force on each of the electrode of the Lablet from the electrodes of all the other Lablet. Then, pair torque will be calculated taking cross product of position wrt. to COM of the Lablet. So, we define the cross product as :

```
crossProduct[{a1_, b1_}, {a2_, b2_}] := a1 b2 + a2 b1
```

```
torque[{x_, y_}, {xc_, yc_}, {x0_, y0_, θ0_}] :=
  crossProduct[{x - xc, y - yc}, FieldLablet[{x, y}, {x0, y0, θ0}]]
```

```
pairTorque[{x_, y_, θ_}, {x0_, y0_, θ0_}] =
    Total@(torque[#, {x, y}, {x0, y0, θ0}] & /@genPositions[x, y, θ]);
```

```
{pairTorque[{-1, -1, 3π/4}, {5, 6, 2π/3}], pairTorque[{5, 6, 2π/3}, {-1, -1, 3π/4}]} // N
```

$\{-1.83322\times10^{-8}, -1.89929\times10^{-8}\}$

Coupled governing rotational and translational equations of motion

Governing rotational and translational equations of motion are given by :

```
rules = {x1 → x1[t], x2 → x2[t], y1 → y1[t], y2 → y2[t], θ1 → θ1[t], θ2 → θ2[t]};
```

```
pairForce[{x1_, y1_, θ1_}, {x2_, y2_, θ2_}] =
    Total@(FieldLablet[#, {x1, y1, θ1}] & /@genPositions[x2, y2, θ2]);
```

```
{Fxp1, Fyp1} = pairForce[{x1, y1, θ1}, {x2, y2, θ2}] /. rules;
```

```
{Fxp2, Fyp2} = pairForce[{x2, y2, θ2}, {x1, y1, θ1}] /. rules;
```

```
τp1 = pairTorque[{x1, y1, θ1}, {x2, y2, θ2}] /. rules;
```

```
τp2 = pairTorque[{x2, y2, θ2}, {x1, y1, θ1}] /. rules;
```

```
govEquations[{m1 , m2 }, {I1, I2}, {p, q, n}] = {Fxp1 == m1 x1''[t], Fyp1 == m1 y1''[t],
     Fxp2 == m1 x2''[t], Fyp2 == m2 y2''[t], τp1 == I1 θ1''[t], τp2 == I2 θ2''[t]};
```

```
initVelocities =
    {x1'[0] == 0.05, y1'[0] == 0, x2'[0] == 0, y2'[0] == 0, θ1'[0] == 0, θ2'[0] == 0 };
```

```
initPositions = {x1[0] == 1, y1[0] == 1, x2[0] == 5, y2[0] == 6, θ1[0] == π/3, θ2[0] == π/4};
```

```
initConditions = {initVelocities, initPositions} // Flatten;
```

```
totalEquations[{m1_ , m2_ }, {I1_, I2_}, {p_, q_, n_}] =
    {govEquations[{m1 , m2 }, {I1, I2}, {p, q, n}], initConditions} // Flatten;
```

```
eqnsFinal = totalEquations[{1, 1}, {1, 1}, {0.1, 10, 1}] // Flatten;
```

```
sol = NDSolve[eqnsFinal, {x1[t], y1[t], x2[t], y2[t], θ1[t], θ2[t]}, {t, 0, 1000}]
```

```
{{x1[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t],
  y1[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t],
  x2[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t],
  y2[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t],
  θ1[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t],
  θ2[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t]}}
```

```
Plot[Evaluate[({θ1[t], θ2[t]} /. sol)], {t, 0, 90}, PlotStyle → Thick]
```



```
Show[ParametricPlot [({x1[t], y1[t]} /. sol), {t, 0, 90}, PlotStyle → {Blue, Thick}],
  ParametricPlot [({x2[t], y2[t]} /. sol), {t, 0, 90}, PlotStyle → {Red, Thick}],
  PlotRange → All]
```




```
plotLablet2D[pts_] := {Yellow, Polygon[pts], Black, PointSize[0.01], Point[pts]};

plotLablets[{{x1_, y1_, θ1_}, {x2_, y2_, θ2_}}] :=
    Block[{p1, p2}, p1 = genPositions[x1, y1, θ1];
      p2 = genPositions[x2, y2, θ2];
      Graphics[{LightYellow, Polygon[{{-10, -10}, {10, -10}, {10, 10}, {-10, 10}}],
          plotLablet2D[p1], plotLablet2D[p2]}]];

plotFunc[t_] = plotLablets[
      {{x1[t], y1[t], θ1[t]} /. sol // Flatten, {x2[t], y2[t], θ2[t]} /. sol // Flatten}];
```

```
Animate [plotFunc[t], {t, 1, 90}, AnimationRunning → False]
```